
Rozdział 22

Numeryczne rozwiązywanie cząstkowych

równań różniczkowych

22.1. Równanie przepływu ciepła

W rozdziale 19 przedstawiono prosty przykład zastosowania metody

Eulera do numerycznego rozwiązywania równań różniczkowych. Opi-

sane tam podejście może być zastosowane do rozwiązywania cząstko-

wych równań różniczkowych. Rozważmy znane z termodynamiki rów-

nanie przepływu ciepła, które dla układu jednowymiarowego przyjmuje

postać:

∂u(t, x)
∂t

= c
∂2u(t, x)

∂x2
, (22.1.1)

gdzie: x – unormowana współrzędna przestrzenna 0 ≤ x ≤ 1,

t – czas t > 0.

Przyjęto następujący warunek początkowy oraz warunki brzegowe:

u(0, x) = f(x),
u(t, 0) = α, (22.1.2)

u(t, 1) = β.

Równanie (22.1.1) należy zastąpić równaniem różnicowym postaci:

uk+1
i = uk

i +
c∆t

∆x2
(uk

i+1 − 2uk
i + uk

i−1), (22.1.3)

Po prostym przekształceniu równania (22.1.3) i porównaniu go z

(22.1.1) można rozpoznać jak odpowiednie pochodne zostały zastąpione

przez odpowiednie ilorazy różnicowe. W równaniu (22.1.3) przyjęto wa-

runki początkowe (22.1.2) zapisane poniżej:

u0
i = f(xi),

uk
0 = α, (22.1.4)

uk
n = β,

gdzie: uk
i – odpowiada aproksymacji rozwiązania dla u(k∆t, i∆x).

Zasadę numerycznego rozwiązywania równania (22.3.3) najlepiej bę-

dzie objaśnić wykorzystując siatkę różnicową przedstawioną na rys. 22.1.

Oś pozioma odpowiada współrzędnej przestrzennej x, natomiast oś pio-

x

t

i,k

i-1 i i+1

k

k+1

k-1

Rys. 22.1. Siatka różnicowa

nowa współrzędnej czasowej t. Ponieważ proces rozwiązywania ma cha-

rakter dyskretny obie osie podzielono na odcinki o długości ∆x i ∆t.
Węzły tak skonstruowanej siatki można ponumerować przy użyciu pary

liczb i, k. Korzystając z warunków początkowych (22.1.4) określamy

wartości funkcji u(t, x) = f(x) w pierwszej warstwie siatki, to jest dla

współrzędnej czasowej t = 0, symbolicznie zapiszemy to w postaci u1
i =

f(xi). W jej punktach skrajnych wpisujemy warunki brzegowe u1
1 = α

i u1
n = β. Obliczenie wartości funkcji u(t, x) w następnej warstwie k = 2

polega na wykorzystaniu zależności (22.1.3), co jest możliwe ponieważ

znamy wszystkie wartości po jej prawej stronie.

Budowę programu rozpoczynamy od ustawienia maksymalnego roz-

miaru stosu, jest to niezbędne jeśli dokonamy gęstego podziału zmien-

nej przestrzennej i czasowej.

stacksize(80000000);

Kolejne polecenie to określenie gęstości podziału zmiennej przestrzen-

nej x, wynika z niego krok ∆x. Jego wartość powiązana z wartością

∆t ma istotne znacznie dla stabilności stosowanego algorytmu. Przyjęty

sposób obliczania wartości ∆t = 1/n2 jest wynikiem przeprowadzonych

eksperymentów numerycznych.

n=50; // podział zmiennej x
dx = 1/n; // krok na siatce x
dt=0.5/n^2; // krok czasowy

Znaczenie następnych dwóch poleceń wyjaśni się jeżeli powrócimy do

równania (22.1.3).

c = 1.0;
lambda =dt/dx^2*c;

Kolejne dwie instrukcje tworzą wektory zawierające dyskretne wartości

zmiennej przestrzennej oraz zmiennej czasowej:

x = (0:dx:1); // wektor x
t = 0; T=(t:dt:1); // wektor T

Podstawowa tablica wewnątrz której będziemy przechowywać wszystkie

wyznaczone wartości funkcji u(t, x) jest tworzona poleceniem zeros(),

jej wymiary muszą być oczywiście zgodne z wymiarami wektorów x i t.

u = zeros(length(x),length(T));

Warunki początkowe i brzegowe definiowane są poleceniami:

alfa=0; // warunek brzeg. lewy
beta=0; // warunek brzeg. prawy
u(:,1)=sin(%pi*x)’; // warunek początkowy
u(1,1)=alfa; u($,1)=beta;

Podstawowa pętla realizująca obliczenia wartości funkcji u(t, x), według

zależności (22.1.3), wykonywana jest wewnątrz pętli while. Zmienna kol
określa numer „warstwy” na której obliczamy wartości funkcji u(t, x).

kol=1;
while t<1 do

t=t+dt;
kol=kol+1;

u(2:$-1,kol) = (1-2*lambda)*u(2:$-1,kol-1) ...
+ lambda*u(3:$,kol-1) ...
+ lambda*u(1:$-2,kol-1);

end

Ostatnia grupa instrukcji służy do graficznej prezentacji zmian strumie-

nia ciepła w funkcji współrzędnej przestrzennej x i czasowej t.

xbasc();
plot3d(x,T,u(:,1:length(T)), ...

theta=20,alpha=65,leg=’x@t@u(t,x)’,flag=[32 2 3]);

22.2. Równanie struny

Pokazany w poprzednim punkcie sposób konstruowania funkcji do

numerycznego rozwiązywania cząstkowych równań różniczkowych może

być względnie łatwo przystosowywany do rozwiązywania różnych typów

równań różniczkowych. Jego adaptacja zostanie pokazana na przykła-

dzie równania opisującego nietłumione drgania struny. Zjawisko roz-

chodzenia się fal opisuje się za pomocą równań różniczkowych cząstko-

wych. Jako przykład weźmy strunę. W każdej chwili czasu t ≥ 0 kształt

struny drgającej przedstawiony jest wykresem pewnej funkcji u(t, x)
która mierzy odchylenie struny od położenia ustalonego. Jeśli znamy

funkcję u(t, x), to znamy ruch drgający struny. Ruch ten jest spowodo-

wany działaniem sił naprężeniowych na każdy jej element. Struna wy-

dłuża się lub kurczy, przy tym zachowana jest zasada Hooke’a o stałości

stosunku wydłużenia do naprężenia. Na podstawie tego można wypro-

wadzić równanie dla funkcji u(t, x)

∂2u(t, x)
∂t2

− c
∂2u(t, x)

∂x2
= 0, (22.2.1)

gdzie: x – unormowana współrzędna przestrzenna 0 ≤ x ≤ 1,

t – czas t > 0.

Przyjęto następujące warunki początkowe i brzegowe:



22.2. Równanie struny 313

u(0, x) = f(x),
∂u(0, x)

∂t
= g(x) (22.2.2)

u(t, 0) = α,

u(t, 1) = β.

Równanie (22.2.1) należy zastąpić równaniem różnicowym postaci

uk+1
i = 2uk

i − uk−1
i +

c∆t2

∆x2
(uk

i+1 − uk
i + uk

i−1). (22.2.3)

Po prostym przekształceniu równania (22.2.3) i porównaniu go z (22.2.1)

można rozpoznać jak odpowiednie pochodne zostały zastąpione przez

odpowiednie ilorazy różnicowe. W równaniu (22.2.3) przyjęto warunki

początkowe i brzegowe (22.2.2) zapisane w sposób pokazany poniżej:

u0
i = f(xi),

uk
0 = α, (22.2.4)

uk
n = β,

gdzie: uk
i – odpowiada aproksymacji rozwiązania dla u(k∆t, i∆x).

Drugi warunek początkowy z (22.2.2) został określony w postaci od-

powiednio zadanych wartości w pierwszych trzech „warstwach” naszej

siatki. Dla uproszczenia przyjęto, że pochodna funkcji u(t,x) jest wielko-

ścią stałą. Program zaczyna się podobnie jak poprzedni od ustawienie

stosu i określenia gęstości podziału zmiennej przestrzennej i czasowej.

Definiowane są również dwa wektory zawierające wartości dyskretne

zmiennej przestrzennej x i czasowej t.

stacksize(80000000);
n=100; // podział zmiennej x
dt=1/n; // krok czasowy
dx = 1/n; // krok na siatce x
c = 1.0; // stałe z równania
lambda = ((dt^2)/(dx^2))*c; // stałe z równania

x = (0:dx:1); // dyskretyzcja x
t = 0;
T=(t:dt:10); // dyskretyzcja t

Podstawowa tablica zawierająca położenie poszczególnych punktów drga-

jącej struny to:

u = zeros(length(x),length(T));

Warunki brzegowe i początkowe to:

alfa=0; // warunek brzeg. lewy
beta=0; // warunek brzeg. prawy
u(:,1)=sin(%pi*x)’; // warunek początkowy
u(1,1)=alfa; u($,1)=beta;
g=1;

Drugi warunek początkowym (zależność 22.2.2) został zapisany w po-

staci ciągu instrukcji warunkowych:

if (g==1) then
u(2:$-1,2) = u(1:$-2,1);
elseif (g==-1) then
u(2:$-1,2) = u(3:$,1);
else
u(2:$-1,2) = 0.5*(u(1:$-2,1)+u(3:$,1));
end
u(1,2) = alfa; u($,2) = beta;
u(1,3) = alfa; u($,3) = beta;

Kolejna grupa poleceń umożliwia wizualną kontrolę zgodności warun-

ków początkowych i brzegowych:

xbasc(); plot2d(x,u(:,1),rect=[0,-1,1,1]);
xbasc(); plot2d(x,u(:,2),rect=[0,-1,1,1]);

Główna pętla obliczeniowa skonstruowana jest podobnie jak ta dla za-

gadnienia przepływu ciepła.

kol=2;
while t<10 do
t=t+dt;
kol=kol+1;
for j=2:n do
u(j,kol) = (2-2*lambda)*u(j,kol-1) ...

- u(j,kol-2) ...

+ lambda*u(j+1,kol-1) ...
+ lambda*u(j-1,kol-1);

end

Polecenie plot2d() wewnątrz pętli while umożliwia obserwację drgają-

cej struny na ekranie monitora.

xbasc(); plot2d(x,u(:,kol),rect=[0,-1,1,1]);
xpause(5000);
end

Program kończy się kreśleniem wychyleń struny (rys. 22.2) dla wszyst-

kich analizowanych chwil czasowych.

xbasc();
plot3d(x,T,u(:,1:length(T)),...
theta=20,alpha=65,leg=’x@t@u(t,x)’,flag=[32 2 3]);

Rys. 22.2. Drgania struny

22.3. Nieliniowe cząstkowe równanie różniczkowe

sinus-Gordona

Własności wahadła matematycznego są powszechnie znane. W ni-

niejszym punkcie chcemy skupić naszą uwagę na numerycznym rozwią-

zaniu równania opisującego bardzo złożony układu wielu sprzężonych

ze sobą wahadeł matematycznych. Uzyskane rozwiązanie ma ogromne

znaczenie teoretyczne ale i praktyczne, jest wykorzystywane w wielu

działach fizyki i jej zastosowań [6]. Nie bez znaczenia jest również mate-

matyczne piękno uzyskiwanych rozwiązań. Na rysunku 22.3 pokazano

prosty przykład takiego układu. Składa się on z wału skrętnego oraz

osadzonych w stałych odległościach od siebie wahadeł matematycz-

nych. Rozważmy dalej równowagę układu trzech wahadeł (rys. 22.4)

Rys. 22.3. Łańcuch wahadeł matematycznych

[26]. Równanie równowagi k-tego wahadła przyjmuje postać:

l(∆mlεk) = −lFk + Mk,k+1 + Mk,k−1, (22.3.1)

Pamiętając, że oddziaływanie pomiędzy dowolnymi trzema sąsiednimi

wahadłami jest proporcjonalne do różnic ich wychyleń kątowych, tj.

Mk,k±1 = α(ϕk±1 − ϕk)/∆, gdzie współczynnik α charakteryzuje sprę-

żyste własności materiału (dla drgań skrętnych) oraz, że składowa siły

ciężkości działająca na k-te wahadło, to Fk = ∆mgsin(ϕk), gdzie ∆m

Rys. 22.4. Trzy sąsiednie wahadła

jest masą punktu materialnego na końcu wahadła, m gęstością liniową

, równanie (22.3.1) przyjmuje postać:

∆ml2εk = −∆mglsin(ϕk) + α(ϕk+1 + ϕk−1 − 2ϕk)/∆, (22.3.2)

dla k = 1, 2, . . . .

Ponieważ przyspieszenie kątowe εk jest drugą pochodną wychylenia

względem czasu, ten układ równań można zapisać jeszcze inaczej:

l

g

d2ϕk(t)
dt2

= −sin(ϕk(t)) +
α

mgl

ϕk+1(t) + ϕk−1(t) − 2ϕk(t)
∆2

. (22.3.3)

Równanie (22.3.3) to układ równań różniczkowo-różnicowych, opisujący

własności ciągu wahadeł matematycznych oddziaływujących ze sobą

według najprostszego sposobu: oddziaływanie dotyczy jedynie sąsied-

nich wahadeł i jest liniowe względem różnicy wychyleń. Jeżeli współ-

czynnik α będzie zmierzał do zera, równanie (22.3.3) przyjmuje znaną

postać równania opisującego ruch swobodny oscylatora harmonicznego.

Jeśli natomiast będziemy zmniejszać masę wahadeł i równocześnie zmniej-

szać odległość pomiędzy nimi, to ostatni człon równania (22.3.3) zmierza

do
∂2ϕ
∂x2 , a całe równanie przyjmuje postać

l

g

d2ϕk(t)
dt2

= −sin(ϕk(t)) +
α

mgl

∂2ϕ(x, t)
∂x2

, (22.3.4)

gdzie poszukiwana funkcja ϕ(x, t) zależy teraz od współrzędnej x mie-

rzonej wzdłuż łańcucha wahadeł a nie od numeru wahadła. Zależy oczy-

wiście od czasu t.

Dokonując podstawień:

τ =
√

g
l t, ξ =

√
mgl
α x,

otrzymujemy ostatecznie

∂2ϕ(ξ, τ)
∂ξ2

− ∂2ϕ(ξ, τ)
∂τ2

= sin(ϕ(ξ, t)). (22.3.5)

Otrzymane nieliniowe równanie różniczkowe cząstkowe nosi nazwę

równania sinus-Gordona. Równanie to ma nieskończenie wiele rozwią-

zań. Jednak ich konstrukcja analityczna jest często bardzo trudna. Po-

łowa z nich daje się jednak przedstawić przez funkcje elementarne.

Jedno z tych rozwiązań nosi nazwę solitonu i ma bardzo ciekawe wła-

sności.

Zadanie znalezienia analitycznego rozwiązania równania (22.3.5) przy

użyciu standardowych poleceń dostępnych w Maximie jest skazane na

niepowodzenie. Pozostaje więc próba znalezienia rozwiązania nume-

rycznego. Pierwszym krokiem jest oczywiście zamiana cząstkowego

równania sinus Gordona na równanie różnicowe. W literaturze można

znaleźć wiele różnych schematów różnicowych [6] pozwalających roz-

wiązać równanie sinus Gordona, do dalszej analizy wybierzemy jeden z

nich

uk+1
i = −uk−1

i +
(

∆x

∆t

)2

(uk
i+1 +uk

i−1)+2(1−
(

∆x

∆t

)2

)uk
i −

(
∆x

∆t

)
sin(uk

i ),

(22.3.6)

gdzie
(

∆x
∆t

)2 ≈ 0.95. Rozwiązywanie równania różnicowego (22.3.6) wy-

maga zdefiniowania warunków początkowych i brzegowych w naszym

przypadku połączonych ze sobą:

u(0, x) = sin(x),
∂u(0, x)

∂t
= g(x). (22.3.7)

Przyjęty sposób implementacji równania różniczkowego identyczny z

tym opisanym w punkcie 22.2.



22.3. Nieliniowe cząstkowe równanie różniczkowe sinus-Gordona 319

stacksize(80000000);
n=500; // podział zmiennej x
dt=100/(n*90); // krok czasowy
dx = 1/n; // krok na siatce x
r = dx/dt; // stale z równania
x = (0:dx:1); // dyskretyzcja x
t = 0;
T=(t:dt:2); // dyskretyzcja t
u = zeros(length(x),length(T));
alfa=0.; // warunek brzeg. lewy
beta=0; // warunek brzeg. prawy
u(:,1)=sin(%pi*x)’; // warunek początkowy
u(1,1)=alfa;
u($,1)=beta;
g=0;
if (g==1) then

u(2:$-1,2) = u(1:$-2,1);
elseif (g==-1) then

u(2:$-1,2) = u(3:$,1);
else
u(2:$-1,2) = 0.5*(u(1:$-2,1)+u(3:$,1));
end
u(1,2) = alfa; u($,2) = beta;
u(1,3) = alfa; u($,3) = beta;
kol=2;
while t<2

t=t+dt;
kol=kol+1;
for j=2:n do

u(j,kol)=-u(j,kol-2)+...
r^2*(u(j+1,kol-1)+u(j-1,kol-1))..
+2*(1-r^2)*u(j,kol-1)-dx*sin(u(j,kol-1));

end
xbasc(); plot2d(x,u(:,kol),rect=[0,-2,1,2]);
xpause(5000);

end
xbasc();
plot3d(x,T,u(:,1:length(T)),theta=20,alpha=65,...

leg=’x@t@u(t,x)’,flag=[32 2 3]);

Po uruchomieniu powyższego programu możemy obserwować jedno z

rozwiązań naszego równania, zwane breather („oddychacz”) odpowiada

ono parze rozwiązań soliton i antysoliton (rys. 22.5), które w naszym

przypadku pulsują nie zmieniając swojego położenia.

Rys. 22.5. Para soliton i antysoliton, nazywana czasami oddychacz (breather)

